

Appendice A. Breve guida al simulatore elettrico SPICE

Lo scopo dei simulatori elettrici è l'analisi del comportamento di reti elettriche rappresentate da insiemi di nodi ai quali sono connessi i terminali dei dispositivi. I simulatori elettrici devono consentire la previsione dell'andamento di tutte le grandezze elettriche (tensioni e correnti) del circuito nel massimo dettaglio possibile. I componenti vengono rappresentati da equazioni che legano, nel modo più possibile aderente alla realtà, le varie grandezze elettriche (correnti e tensioni) presenti ai terminali dei componenti stessi. Il simulatore elettrico si distingue quindi dai simulatori logici, che non trattano tensioni e correnti ma solo gli stati logici che esse codificano e dai simulatori misti, che rappresentano i componenti in modo estremamente semplificato. I simulatori elettrici costituiscono uno strumento indispensabile per il progetto di circuiti analogici integrati e per l'analisi delle prestazioni di ogni nuova cella digitale che viene sviluppata.

Capostipite dei simulatori elettrici è il programma SPICE (*Simulation Program with Integrated Circuit Emphasis*), sviluppato nella sua prima versione all'inizio degli anni '70 presso l'università di Berkeley.


SPICE utilizza diverse strategie di risoluzione della rete a seconda del tipo di analisi che deve effettuare. Nel caso di calcolo del punto di lavoro, operazione preliminare a tutte le altre analisi, vengono risolte le equazioni della rete mediante il metodo iterativo di Newton Raphson, opportunamente modificato per migliorare la convergenza con le tipiche equazioni non lineari introdotte dai componenti a semiconduttore. Nel caso di analisi in transitorio vengono integrate le equazioni differenziali che regolano l'evoluzione delle varie variabili di stato (cariche sulle capacità e dei flussi magnetici nelle induttanze). L'analisi nel dominio della frequenza viene invece effettuata sul circuito linearizzato attorno al punto di riposo e prevede quindi la soluzione di sistemi lineari a coefficienti complessi.

La rete elettrica, gli stimoli applicati ad essa, il tipo di analisi da effettuare, i risultati da presentare in uscita ed altri parametri della simulazione (opzioni) vengono specificati in un file di ingresso testuale (ASCII), il cui formato verrà descritto nel seguito di questa appendice. Grazie a comandi opportuni inseriti nel file di input, si può comandare l'inclusione di altri file o indicare al simulatore quali file di libreria devono essere scanditi per la ricerca delle eventuali informazioni mancanti (tipicamente modelli di dispositivi o descrizione di sottocircuiti). Il simulatore produce sempre un file di uscita (tipicamente di estensione .out), sempre in formato ASCII, contenente i risultati della simulazione oppure, nell'eventualità di errori, la descrizione di questi ultimi.

Si fa osservare che la maggior parte dei simulatori elettrici più diffusi (SPICE2 e SPICE3 di Berkeley, HSPICE, ELDO, Spectre, PSPICE), utilizzano formati praticamente coincidenti per i file testuali di ingresso e uscita. Come risultato della simulazione si possono generalmente produrre file binari contenenti le funzioni di uscita in un formato adeguato per la rappresentazione grafica che viene effettuata da programmi distinti dal simulatore stesso. I formati di questi file dipendono esclusivamente dal tipo di visualizzatore supportato dal simulatore.

In questa guida riassuntiva viene descritta la sintassi dei comandi e delle dichiarazioni più comuni che possono essere introdotti nel file di input. Nella pagina seguente vengono mostrati gli elementi principali che costituiscono la struttura del file di input.

Struttura semplificata di un file di ingresso in formato SPICE



Convenzioni

In questa breve guida a SPICE verrà seguita la convenzione di indicare tra parentesi angolari “< >” i parametri variabili; e tra parentesi quadre [] gli argomenti opzionali. La sintassi di SPICE non fa distinzione tra caratteri maiuscoli e minuscoli. Accanto al formato generico, supportato universalmente da tutti i simulatori della famiglia di SPICE, verranno indicate quelle variazioni che caratterizzano due simulatori di interesse in ambito didattico: PSPICE incluso nell’ambiente di progettazione Orcad per schede a circuito stampato, e T-SPICE, che fa parte dell’ambiente di progettazione di circuiti integrati della Tanner.

Fattori moltiplicativi.

La maggior parte dei parametri indicati nel file di input sono espressi in unità del sistema internazionale. Vi sono alcune eccezioni, quali il parametro mobilità (UO) della “model” del MOSFET e i drogaggi, legati alla consuetudine di utilizzare unità differenti. I valori dei parametri possono essere indicati nei più comuni formati usati per i numeri, con eventuale aggiunta dell’esponente. Esempi: 1.26, 0.001, 1.7E-6. Per migliorare la leggibilità si possono usare i seguenti moltiplicatori:

F	10 ⁻¹⁵	(femto)	K	10 ³	(kilo)
P	10 ⁻¹²	(pico)	MEG	10 ⁶	(mega)
N	10 ⁻⁹	(nano)	G	10 ⁹	(giga)
U	10 ⁻⁶	(micro)	T	10 ¹²	(tera)
M	10 ⁻³	(milli)			

Componenti passivi.

Resistori

Sintassi:

```
R<nome> <N+> <N-> <valore>
```

Esempio:

```
R10 1 3 10k
```

***** Espansione PSPICE *****

```
R<nome> <N+> <N-> [nome_model] <valore>
```

La model per un resistore assume la forma:

```
.model <nome_model> RES (R=<moltiplicatore> TC1=<tc1> TC2=<tc2>)
```

La resistenza risulta:

$$\text{Resistenza} = R \cdot \text{valore} \cdot \left(1 + TC1(T - T_{nom}) + TC2(T - T_{nom})^2\right).$$

***** Espansione T-Spice *****

```
R<nome> <N+> <N-> <valore> [TC=A, [B]]
```

con A che fa le veci di TC1 e B di TC2 in PSPICE.

Induttanze

Sintassi:

L<nome> <N+> <N-> <valore> [IC = <corrente iniziale>]

Nota: l'opzione IC che fissa la condizione iniziale (corrente) per il transitorio non è valida in T-SPICE.

Condensatori

Sintassi:

C<nome> <N+> <N-> <valore>

Esempio:


Cload 23 12 100n

***** Espansione PSPICE *****
C<nome> <N+> <N-> [<nome_model>] <valore> [IC=<Tensione iniziale>]
Tipo della model: CAP
Parametri della model: C, VC1, VC2, TC1,TC2.
Espressione della capacità:

$$Capacità = C \cdot valore \cdot (1 + VC1 \cdot V + VC2 \cdot V^2) \cdot (1 + TC1(T - T_{nom}) + TC2(T - T_{nom})^2)$$

V è la tensione ai capi della capacità presa con polarità dal nodo N+ al nodo N-. Il parametro IC fissa il valore iniziale della tensione per le simulazioni in transitorio.

Mutue Induttanze



Date due induttanze già definite con l'apposita sintassi, si può introdurre un accoppiamento tra le due nel seguente modo:

K<nome> <induttanza 1> <induttanza 2> <valore di k>


La figura chiarisce la relazione tra l'ordine con cui sono introdotti i nodi nella dichiarazione delle due induttanze e i "pallini" convenzionali che indicano la polarità delle tensioni indotte. Si riporta la relazione tra la mutua induttanza e il coefficiente K:

$$M = K \cdot \sqrt{L_1 L_2}$$

K deve essere compreso tra 0 e 1.

Generatori indipendenti

Esistono due tipi di generatori indipendenti: generatori di corrente e generatori di tensione. La figura mostra la convenzione sulle polarità dei generatori.



Generatori indipendenti di tensione

Sintassi:

V<nome> <N+> <N-> DC <valore in d.c.> [AC <ampiezza> <fase in gradi>]
[<funzione transitorio>]

Generatori indipendenti di corrente

Sintassi:

I <nome> <N+> <N-> DC <valore in d.c.> [AC <ampiezza> <fase in gradi>]
[<funzione transitorio>]

Significato dei campi che caratterizzano i generatori indipendenti:

DC: Fissa il valore del generatore per il calcolo del punto di riposo.

AC: Fissa i valori di ampiezza e fase da usare nell'analisi nel dominio della frequenza. Si tenga presente che in tale analisi tutti i generatori che hanno una specifica AC vengono accesi contemporaneamente (con i valori di ampiezza e fase specificati). La frequenza dei generatori viene fissata uguale per tutti e variata su un intervallo prestabilito dai parametri dell'analisi AC (si veda il paragrafo sul tipo di analisi). I generatori che non possiedono un campo AC sono spenti durante la suddetta analisi. Essi contribuiscono comunque a fissare il punto di lavoro attorno al quale il circuito viene linearizzato prima di effettuare l'analisi nel dominio della frequenza (analisi AC).

Funzione transitorio: specifica la forma d'onda che produce il generatore durante l'analisi in transitorio. I generatori che non possiedono questo campo rimangono fissati al valore specificato dal campo DC. La sintassi delle funzioni usate più comunemente è descritta nella pagina seguente.


Sintassi di alcune funzioni che definiscono il comportamento dei generatori durante l'analisi in transitorio.

1) Sinusoidale con eventuale smorzamento.

Sintassi (varia molto a seconda del tipo di implementazione del simulatore):

SIN(<V₀> <A> <f> <t_R> <d> <φ>)


$$f(t) = \begin{cases} V_0 + A \cdot e^{-d \cdot (t-t_R)} \sin(2\pi f(t-t_R) + \varphi / 360) & \text{per } t > t_R \\ V_0 & \text{per } t < t_R \end{cases}$$



2) Lineare a tratti.

Sintassi:

PWL(<t1> <v1> <t2> <v2> <t3> <v3>)




se $t_1 \neq 0$ per $t=0$ vale il valore in DC (come in figura).

3) Sequenza di impulsi.

Sintassi:

PULSE(<val. iniziale> <val. impulsato> <t. ritardo> <t. salita> <t. discesa> <durata> <periodo>)



Nota: in T-SPICE non è ammesso specificare un valore DC e contemporaneamente una funzione transitorio. Nel caso sia specificata la funzione transitorio il valore in continua è dato dal valore della funzione transitorio per $t=0$.

ESEMPI:


Vdd 11 0 DC 5

Vin 23 12 DC 0 AC 1 180

I0 4 0 DC 1m PWL(0 1m 1u 1m 1.1u 0 2u 0)

Generatori dipendenti

La figura seguente illustra i quattro possibili generatori dipendenti in SPICE. Si osservi che per i generatori comandati da una corrente (H e F), la corrente di comando deve scorrere in un generatore indipendente di tensione. Per far riferimento ad una corrente che scorre in un ramo in cui non sono presenti generatori indipendenti di tensione, si può introdurre in serie al ramo stesso un generatore opportuno di tensione nulla che non perturba il circuito e, nello stesso tempo, funziona da “amperometro”.



Di tensione controllato in tensione:

E<nome> <N+> <N-> <Nc+> <Nc-> <e>

Di corrente controllato in tensione:

G<nome> <N+> <N-> <Nc+> <Nc-> <g>

Di corrente controllato in corrente

F<nome> <N+> <N-> <sorgente> <f>

Di tensione controllato in corrente

H<nome> <N+> <N-> <sorgente> <h>

Nota: con “sorgente” si è indicato il generatore di tensione usato per poter riferire la corrente di comando.

Esempi di generatori dipendenti

```
E2 1 2 3 44 -1
Gms 2 33 21 3 1m
Fhf1 1 2 Vre 300
H23 11 2 Vin 1e3
```

Componenti a semiconduttore

Nella dichiarazione di un qualsiasi dispositivo a semiconduttore si deve fare sempre riferimento alla model, ovvero a quella parte di file di input (o di libreria) dove vengono specificati parametri del dispositivo. Per la sintassi delle model si rimanda all'appendice 2.

Diodi

D<nome> <N+> <N-> <nome model> [<area>]

Il nodo N+ è l'anodo, mentre n- è il catodo.

Esempio di dichiarazione di un diodo.

```
D22 1 8 mio_diodo 2
```

Nota: il parametro area agisce come un parametro adimensionale di scala su alcuni parametri specificati nella model del diodo. Per esempio la corrente di saturazione IS viene moltiplicata per l'area, le resistenze serie vengono divise per l'area.

Transistori Bipolari

Q<nome> <collettore> <base> <emettitore> [<substrato>] <nome model>
<area>

Esempio di dichiarazione di BJT (si è omessa l'area che per default vale 1)

```
Q11 2 5 6 78 miobjt
```

Nota: collettore, base, etc. sono i nodi a cui sono collegati i corrispondenti terminali dei transistori.

Per quanto riguarda l'area vale quanto detto per il diodo.

Transistori MOS.

M<nome> <drain> <gate> <source> <substrato> <nome model> [L=<l>]
[W=<w>] [...altri parametri ..]

Esempio di definizione di MOSFET:

```
M2 2 5 11 3 mio_mos L=1u W=2u AD=2p AS=8.5p
```

La lista di parametri è costituita da definizioni di tipo: Parola chiave = <valore> come mostrato per i due parametri L e W. I parametri specificabili sono i seguenti:

Parametro	Significato	Default PSPICE	Default T-SPICE
L	Lunghezza del canale	defl	defl
W	Larghezza del canale	defw	defw
AD	Area di drain	defad	LW oppure defad
PD	Perimetro di drain	0	2(L+W)
AS	Area di source	defas	LW oppure defas
PS	Perimetro di source	0	2(L+W)
NRD	Numero di quadri della diffusione di drain	1	defnrd
NRS	Numero di quadri della diffusione di source	1	defnrs

Transistori JFET.

J<nome> <drain> <gate> <source> <nome_model> [<area>]

Esempio di dichiarazione di JFET:

J75 12 34 11 mio_jfet

Sottocircuiti

I sottocircuiti corrispondono ai moduli (o celle) dello schematic editor. L'uso dei sottocircuiti consente di velocizzare la scrittura della netlist quando una stessa parte di un circuito complessa è ripetuta molte volte (per esempio un operazionale, una porta logica, un flip-flop etc. etc.). Un altro vantaggio inerente l'uso di sottocircuiti è la maggior leggibilità della netlist, con conseguente maggior facilità diagnostica.

Occorre rammentare che SPICE, prima di risolvere la netlist, la espande sostituendo a ciascun sottocircuito il suo contenuto. Se sono presenti più livelli di gerarchia questo processo arriva fino a generare una netlist in cui compaiono solo componenti (celle primitive). Questa operazione viene indicata come "flattening".

Definizione:

```
.SUBCKT <nome_sottocircuito> <lista ordinata di nodi formali>
.....
elementi della rete costituente il sottocircuito
.....
.ENDS
```

Istanza:

```
X<nome> <lista ordinata di nodi > <nome_sottocircuito>
```

Esempio:

Definizione del sottocircuito:

```
.SUBCKT PASSA_BASSO 1 2
R1 1 2 100K
C1 2 0
.ENDS
```

Istanze del sottocircuito (2 passa basso in cascata)

```
X3 44 55 PASSA_BASSO
X4 55 66 PASSA_BASSO
```

Note sull'uso dei sottocircuiti.

- All'interno della rete possono essere introdotti nodi ausiliari non visibili all'esterno.
- Il nodo "0" (nodo di riferimento, denominato anche "gnd" nei programmi che accettano nodi letterali) è sempre un nodo globale, ovvero è visibile ed è lo stesso all'interno di tutti i sottocircuiti. Non occorre quindi passarlo nei nodi formali e non può essere usato come nodo locale (ovvero visibile solo nel sottocircuito).
- In alcune versioni del simulatore è possibile introdurre altri nodi globali (per esempio i *rail* di alimentazione), in modo da migliorare la leggibilità della *netlist*. Si noti l'analogia con quanto accade nello schematic editor.
- Ovviamente un sottocircuito può non contenere connessioni interne a nodi globali.
- La corrispondenza tra nodi formali (nella definizione del sottocircuito) e nodi del circuito utilizzatore (nell'istanza) è stabilita esclusivamente dall'ordine.
- Un sottocircuito può contenere istanze di altri sottocircuiti (struttura gerarchica).
- I le grandezze elettriche relative a componenti e nodi interni ai sottocircuiti possono essere sempre rintracciati, indicandoli con la seguente sintassi:
<Nome istanza>.<grandezza elettrica>. Esempio: X27.V(6) (tensione del nodo 6 interno all'istanza di sottocircuito denominata X27).

Comandi di Spice

I comandi di SPICE si distinguono dalle istanze dei dispositivi che costituiscono la netlist attraverso il punto con il quale iniziano le relative parole chiave. Sono già stati introdotti i comandi relativi alla definizione di un sottocircuito all'introduzione della model dei dispositivi. In questa sezione verrà completata la descrizione dei comandi più comuni.

Scelta del tipo di analisi

Determinazione del punto di riposo.

Sintassi:

```
.OP
```

il comando .OP ha una funzione che dipende dal tipo di Spice che si sta utilizzando:

-) in T-SPICE comanda effettivamente il calcolo del punto di riposo che viene effettuato considerando come note le variabili (tensioni e correnti) fissate con il comando .IC. I valori specificati con .NODESET vengono utilizzati per fornire l'inizio del procedimento iterativo. Per riportare i risultati nel file .out occorre specificare nel comando ".print DC" le variabili di cui si desidera conoscere il valore. Sempre in T-Spice, per avere in uscita le informazioni sui parametri di piccolo segnale dei dispositivi attivi occorre aggiungere una riga di tipo:

```
.ACMODEL { * }
```

-) in PSPICE il punto di riposo viene calcolato sempre e .OP fa sì che vengano riportate nel file .out informazioni dettagliate sul punto di riposo (parametri dei circuiti linearizzati) che altrimenti sarebbero limitate alle sole tensioni nodali.

Nel calcolo del punto di riposo i generatori indipendenti vengono posti al loro valore in continua. Anche qui c'è una differenza tra T-SPICE e PSPICE: in T-SPICE nel caso un generatore abbia specificata una forma d'onda per l'analisi in transitorio viene usato come valore in continua il valore assunto dalla forma d'onda per $t=0$. Nel caso di PSPICE è invece possibile specificare un valore in continua (parola chiave DC) indipendente dalla funzione usata per il transitorio.

Scansione in continua (“sweep d.c.”).

Sintassi:

```
.DC V<nome1> <Vi1> <Vf1> <Vstep1> [V<nome2> <Vi2> <Vf2> <Vstep2>]
```

Il generatore V<nome1> viene incrementato da Vi1 a Vf1 con passo pari a Vstep1. Per ogni valore del generatore V<nome1> viene effettuato il calcolo del punto di riposo (analisi in continua). Se è presente anche la parte opzionale il ciclo viene ripetuto incrementando il generatore V<nome2> da Vi2 a Vf2 con passo Vstep2. In pratica alla fine di ogni ciclo del generatore 1 viene incrementato di un passo il generatore 2. Tutti gli altri generatori indipendenti sono mantenuti fissi al loro valore in continua.

Analisi “AC” (analisi nel dominio della in frequenza).

Sintassi:

```
.AC <suddivisione > <numero_di_punti> <freq. iniz.> <freq. finale>
```

Con l'analisi AC viene linearizzato il circuito attorno al punto di riposo (stabilito dal valore DC dei generatori) e su di esso vengono fatti agire contemporaneamente tutti i generatori indipendenti che hanno specificati i dati per l'analisi AC. Gli altri generatori indipendenti sono cortocircuitati (V) oppure aperti (I) nel circuito linearizzato. Per ogni valore di frequenza richiesto viene calcolato il regime sinusoidale e i risultati sono fasori delle tensioni o correnti. Ovviamente anche il circuito linearizzato è composto di impedenze complesse che vengono ricalcolate per ogni valore di frequenza a cui si effettua l'analisi. Se ad agire è un solo generatore, di ampiezza AC unitaria e fase nulla, le tensioni (o correnti) fornite dall'analisi AC coincidono, in modulo e fase, con la risposta in frequenza rispetto al dato generatore.

Il parametro **suddivisione** può assumere i tre valori: **LIN**, **DEC** e **OCT** e fissa la suddivisione dell'asse delle frequenze in lineare, a decadi o a ottave.

Il numero dei punti è il numero totale di punti sull'asse delle frequenze nel caso **LIN** oppure è il numero di punti per decade o per ottava nei casi **DEC** o **OCT**.

Per esempio la dichiarazione:

```
.AC DEC 9 1 1k
```

Comanda l'esecuzione del calcolo del regime sinusoidale (sul circuito linearizzato!) alle frequenze: 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 20, 30, 40600, 700, 800, 900, 1000.

Funzione di trasferimento in continua.

Sintassi:

```
.TF <variabili dipendendenti> <sorgente indipendente>
```

.TF consente il calcolo della funzione di trasferimento di piccolo segnale (si opera sul circuito linearizzato), in continua, tra un generatore indipendente specificato nel parametro “sorgente indipendente” e una serie di variabili (correnti o tensioni). In sostanza l'analisi è

equivalente a valutare il limite della funzione di trasferimento per la frequenza che tende a zero. Per la sintassi da utilizzare per specificare le variabili dipendenti si veda più avanti il comando “PRINT”.

PSPICE consente una sola variabile dipendente per comando .TF e non necessita di comando ulteriore (es. .PRINT) per produrre il risultato sul file di output.

T-SPICE consente più variabili ma necessita che si specifichi un comando “.PRINT DC <variabili dipendenti>”.

Analisi in transitorio.

Viene effettuata un’analisi di transitorio facendo agire i generatori indipendenti per i quali è stata specificata una forma d’onda. Gli altri generatori indipendenti vengono mantenuti al valore in continua. Il circuito su cui viene calcolato il transitorio non viene linearizzato ma vengono mantenute tutte le non linearità dei componenti.

La sintassi e il significato dei parametri è piuttosto differente tra PSPICE e TSPICE e conviene quindi mostrarle separatamente.

Sintassi PSPICE (comune alla maggioranza delle implementazioni di SPICE)

```
.TRAN <step di stampa> <tempo finale> [<inizio stampa>  
<step interno massimo>] [UIC]
```

L’analisi inizia sempre dall’istante $t=0$ e finisce per il tempo specificato con <tempo finale>. Si può evitare di riportare nei risultati quello che accade prima di un certo istante specificando il parametro opzionale “inizio stampa”.

È importante chiarire il significato del parametro “step interno massimo”. SPICE varia il passo temporale di risoluzione delle equazioni differenziali che regolano il transitorio in modo tale da mantenere la precisione richiesta. In particolare, in presenza di grandezze elettriche che tendono a variare molto più velocemente del passo di risoluzione in vigore, SPICE riduce il passo stesso. Viceversa, se tutte le grandezze elettriche variano lentamente, SPICE “allunga il passo”, in quanto gli è possibile prevedere correttamente il nuovo stato del circuito senza dover effettuare step intermedi. Spesso SPICE fissa un passo di risoluzione molto più grande dell’intervallo di stampa. In questo caso tutti i punti intermedi richiesti per la stampa non sono calcolati esattamente da SPICE ma interpolati tra due risultati esatti. L’effetto si manifesta in una eccessiva angolosità delle curve. Specificando il parametro step interno massimo si fissa un valore massimo per il passo di risoluzione delle equazioni. Fissando tale parametro ad un valore pari o inferiore allo step di stampa si può far sì che i punti forniti in uscita siano tutti caratterizzati da una precisione pari a quella stabilita per la risoluzione del circuito. Ovviamente in questo modo si impedisce a SPICE di “allungare il passo”, e ne risulta aumentato il tempo totale di calcolo.

Nota! se si vuole specificare lo <step interno massimo> occorre anche specificare <inizio stampa>.

La parola chiave [UIC] forza il programma a utilizzare le condizioni iniziali specificate con il comando .IC o specificate direttamente nella definizione dei componenti (IC=...). Senza UIC le condizioni iniziali sono quelle valutate con l’analisi del punto di riposo.

Sintassi T-SPICE

```
.TRAN [/<mode>] < step interno massimo > <tempo finale>  
[START=<inizio intervallo di stampa>]
```

In questo caso viene fissato come parametro obbligatorio il passo di risoluzione delle equazioni (“step interno massimo”) e l’intervallo di tempo simulato (“tempo finale”). Anche qui il transitorio inizia sempre per $t=0$ ma i risultati possono essere forniti in uscita a partire da un <inizio intervallo di stampa>.

Lo step di stampa viene invece fissato pari allo “step interno massimo” e non si hanno quindi i problemi di precisione citati per PSPICE (e tipici anche di molte altre versioni di SPICE, compreso SPICE 2G). L’intervallo di stampa può comunque essere reso differente da “step interno massimo” specificando il parametro “prtdel” in una dichiarazione .OPTIONS.

Il parametro “mode” accetta tre valori (di cui il primo è di default).

dcoppt: viene calcolato inizialmente il punto di riposo usando le condizioni iniziali se specificate nel comando “.IC” e i suggerimenti per la convergenza fissati il comando “.NODESET”.

powerup: viene aiutata la convergenza facendo partire tutti i generatori da zero e facendoli arrivare al valore corretto (per $t=0$) gradualmente. Una volta trovato il punto di riposo per $t=0$ si procede alla simulazione.

preview: serve per controllare se le forme d’onda di ingresso sono corrette: non viene effettuata la simulazione ma vengono riportate in uscita le forme d’onda dei generatori indipendenti specificati nel comando “.PRINT TRAN”

Altri comandi di Spice:

Comandi di universali di output.

```
.PRINT <analisi> <variabili>
```

Il comando .PRINT consente di riportare in modalità di file testuale i risultati di una delle analisi eseguite attraverso gli appositi comandi (per esempio .TRAN). I risultati vengono inseriti nel file di output testuale (.out). Il parametro <analisi> stabilisce che tipo di analisi si vuole riportare come risultato nel file di output.

<analisi> può assumere i valori DC AC TRAN e NOISE che si riferiscono alle rispettive analisi circuitali.

Le variabili sono dichiarate con una sintassi del tipo:

V(<nodo>) per le tensioni.

Per le correnti, si può avere riferimento alle sole correnti nei generatori di tensione e quindi occorre usare la sintassi:

I(V<nome_gen>), dove nome_gen è il nome di un generatore indipendente di tensione.

***** Espansione TSPICE *****

TSPICE consente anche di dichiarare altre variabili tra cui la tensione tra due nodi con la sintassi:

V(<N+>,<N->)

e la corrente entrante in un dispositivo <disp> attraverso il nodo <N> con la sintassi:

I(<disp>,<N>).

Comandi di uscita per post-processor grafici:

TSPICE: è sufficiente il comando .PRINT: il post-processore grafico chiamato WINWED32 legge i file testuali di tipo “.out”.

PSPICE: si può produrre un file di dati (.dat) per il post-processore WPROBE attraverso il comando:

```
. PROBE
```

che fa sì che tutte le tensioni nodali e tutte le correnti in dispositivi attivi e generatori di tensione siano salvate nel file “.dat” Molte implementazioni di spice supportano il comando PROBE. Per restringere l’insieme di variabili salvate nel file si può far seguire al comando PROBE una lista di variabili, come per il comando PRINT.

Utilizzo di informazioni presenti su file esterni:

Consultazioni di file di libreria.

```
.LIB <file di libreria>
```

Questo comando fa sì che venga consultato il file specificato per ricercare definizioni di sottocircuiti o “model” di dispositivi non presenti nel file di input ma utilizzati per la descrizione della rete. I file di libreria sono file di testo che possono contenere solo definizioni di sottocircuiti o “model”. Il programma crea in memoria un file di input espanso nel quale inserisce solo quelle sezioni di librerie che gli servono per la rete in esame (model di dispositivi o definizioni di sottocircuiti citati nel file di input principale).

Inclusione completa di file esterni.

```
.INCLUDE < file di include> (sintassi di TSPICE)
```

```
.INC < file di include> (sintassi di PSPICE)
```

Questi comandi fanno sì che il file specificato venga inserito integralmente nel file di input nel punto esatto in cui tale comando compare. Il file di include può contenere qualsiasi comando o definizione di SPICE. Il file di input viene processato come un unico blocco dopo che sono stati inseriti tutti i file di include.

Specificazione della temperatura.

Sintassi:

```
.TEMP <valore di temperatura>
```

fissa il valore della temperatura alla quale viene effettuata la simulazione. Questo parametro agisce su tutte quelle equazioni costitutive dei dispositivi per le quale è stata inserita una

dipendenza dalla temperatura. Se non è presente il comando “.TEMP” la temperatura è pari all’ “option” TNOM che per default vale 27 °C.

Assegnazione dei parametri globali (options).

Sintassi:

```
.OPTIONS <nome par.>=<valore par.> <nome par.>=<valore par.> .....
```

Esempio:

```
.OPTIONS TNOM=30 DEFW=10u DEFL=2u
```

Il comando .OPTIONS consente di fissare il valore di alcuni parametri che possono influenzare lo svolgimento della simulazione. Il nome e il numero dei parametri cambia a seconda della versione del simulatore. Per ottenere la visualizzazione di tutte le opzioni correnti (nel file .out) si può usare la seguente sintassi:

```
.OPTIONS OPT (sintassi PSPICE)
```

```
.OPTIONS verbose=2 (sintassi T-SPICE)
```

Lista dei parametri di uso più frequente

PSPICE		T-SPICE		Significato del parametro
nome	default	nome	default	
TNOM	27 °C	TNOM	27 °C	temperatura nominale
DEFW	100 µm	DEFW	0	W di default per i MOSFET
DEFL	100 µm	DEFL	0	L di default per i MOSFET
DEFAD	0	DEFAD	0	AD di default per i MOSFET
DEFAS	0	DEFAS	0	AS di default per i MOSFET
GMIN	10 ⁻¹² Ω ⁻¹			Conduttanza messa in parallelo a tutte le
ITL1	40	numnd	100	Max Numero Iterazioni per il calcolo del punto
ITL4	10	numnt	50	Max. Numero di iterazioni per passo per il
ITL5	5000			Max numero di Iterazioni totali per il
LIMPTS	201			Numero massimo di punti in una tabella di
ABSTOL	1 pA	ABSTOL	5 nA	Errore assoluto sulle correnti
RELTOL	10 ⁻¹³	RELTOL	10 ⁻⁷	Errore relativo su correnti e tensioni (solo sulle tensioni in T-SPICE)
VNTOL	1 µV			Errore assoluto sulle tensioni
CHGTOL		CHARGETOL		Errore assoluto sulle cariche nel transitorio.
		RELCHARGETO	10 ⁻³	

Assegnazione di un punto iniziale per aiutare la convergenza in SPICE:

Questo comando consente di suggerire al simulatore il valore della tensione di alcuni nodi. Questi valori sono utilizzati da SPICE come punto di partenza per il metodo iterativo usato per il calcolo del punto di lavoro. Si rammenta che spesso il calcolo del punto di lavoro è il

compito più difficile che deve essere affrontato dal simulatore in quanto le equazioni in gioco sono fortemente non lineari, e, in mancanza di suggerimenti, il simulatore non possiede un valido punto di partenza da cui cominciare le iterazioni. Va precisato anche che tutte le altre analisi si basano sempre sul calcolo di un punto di lavoro come fase preliminare. Ovviamente il risultato del calcolo del punto di lavoro potrà essere completamente diverso da quello suggerito dall'utente mediante il comando NODESET.

Il comando NODESET può risultare utile quando non si riesca a raggiungere la convergenza e la tensione di alcuni nodi sia di facile determinazione da parte dell'utente.

Inoltre, nel caso di circuiti che presentino più stati stabili (per esempio flip-flop), l'uso del comando NODESET può favorire la selezione di una delle varie soluzioni.

Sintassi:

```
.NODESET V(<nodo1>)=<valore1> V(<nodo2>)=<valore2>
```

Assegnazione di condizioni iniziali per il transitorio

L'assegnazione di condizioni iniziali viene utilizzata per fissare la tensione sui condensatori e la corrente nelle induttanze al tempo $t=0$, nell'analisi in transitorio (.TRAN). Non si devono confondere le condizioni iniziali con i valori assegnati con il comando NODESET. Solo nel caso di condizioni iniziali si ha un effettivo significato fisico delle grandezze assegnate e la soluzione (analisi nel dominio del tempo) ne risulta fortemente influenzata. In tutte le analisi diverse dal transitorio le condizioni iniziali sono ignorate (non hanno significato).

L'uso delle condizioni iniziali può essere spesso evitato in quanto, nella maggior parte dei casi di interesse pratico, si sceglie una soluzione a regime, o comunque si inizia ad applicare il segnale (per esempio una sequenza di bit) dopo un intervallo di tempo iniziale introdotto ad arte per far raggiungere al circuito una condizione di regime.

Un caso in cui le condizioni iniziali sono di estrema importanza è quello degli oscillatori. Infatti, non essendo presente nel circuito simulato il rumore che garantisce l'innesco delle oscillazioni in un circuito reale, è possibile spostare dall'equilibrio l'oscillatore assegnando condizioni iniziali diverse da quelle determinate dal punto di lavoro su alcuni elementi reattivi e innescare così l'oscillazione.

Sintassi PSPICE:

```
.IC V(<nodo1>)=<valore1>....
```

Sintassi T-SPICE:

```
.IC V(<nodo1>)=<valore1> . V(<nodo2>,<nodo3>)=<valore>  
I(induttore1)=<valore>....
```

Si noti che PSPICE ha una sintassi apparentemente meno potente ma consente di assegnare le condizioni iniziali anche direttamente sui alcuni componenti (condensatori e induttori). Si veda in proposito la sintassi di dichiarazione dei singoli componenti.

Appendice B. Cenni ai modelli usati da SPICE per gli elementi a semiconduttore.

Tutti gli elementi a semiconduttore presenti nella netlist devono fare riferimento ad una “model”, ovvero ad una sezione del file di input (o di file di libreria), dove vengono specificati i valori dei parametri che definiscono le caratteristiche del componente stesso. Le model hanno il seguente formato:

```
.MODEL <nome_model> <tipo_model> <lista di parametri>
```

Il “nome_model” è un nome convenzionale assegnato dall’utente o, più comunemente, da chi ha compilato la libreria.

Il “tipo_model” specifica di che tipo di componente si tratta (diodo, transistor bipolare etc. etc.). Questa stringa può assumere solo alcuni valori predefiniti stabiliti dal tipo di implementazione di SPICE. I dispositivi di interesse microelettronico presenti in tutte le versioni di SPICE e di simulatori affini sono:

D	diodo
NPN	bjt npn
PNP	bjt pnp
NMOS	n-mosfet
PMOS	p-mosfet
NJF	n-jfet
PJF	p-jfet

La lista di parametri ha un formato del tipo: <nome_parametro>=<valore>. Per esempio: VTO=1.23 (vissa il valore della tensione di soglia di un mosfet). Una distinzione importante riguarda la model dei transistori MOS. In questo caso il primo parametro della lista è del tipo:

```
LEVEL=<numero>
```


Il LEVEL fissa il tipo di modello usato per descrivere il comportamento del mosfet. Per esempio, esistono tre livelli indicati con LEVEL=1, LEVEL=2 e LEVEL=3, supportati da tutte le implementazioni di SPICE e programmi affini. Questi tre livelli rappresentano il comportamento del MOSFET con ordine di complessità crescente. Il LEVEL=1, al quale per semplicità si farà riferimento in questa trattazione, aveva importanza solo i tempi in cui la ridotta potenza di calcolo degli elaboratori poteva sconsigliare l’uso di modelli più complicati.

Attualmente i simulatori supportano un numero elevato di modelli per il mosfet, introdotti per rappresentare quei fenomeni connessi con le dimensioni sub-micrometriche dei moderni dispositivi.

In questa trattazione si farà riferimento ad alcuni parametri validi per i tre livelli base (1,2,3), spesso mantenuti anche per i livelli superiori. In ogni caso, i manuali riportano sempre per ciascun modello supportato, una lista dettagliata dei parametri con il loro significato e l’unità di misura.

Transistori MOS.

Modello per grandi segnali del transistor MOS.



Modello valido per l'n-MOSFET.

S: Source
G: Gate
D: Drain

Equazioni adottate nel LEVEL=1 per la I_{DS}

$$I_{DS} = K_N \frac{L_1}{W_1} \frac{(V_{GS} - V_T)^2}{2} (1 + \lambda V_{DS}) \quad \text{per } V_{DS} > V_{GS} - V_T \quad (\text{A2.1})$$

$$I_{DS} = K_N \frac{L_1}{W_1} V_{DS} \left(V_{GS} - V_T - \frac{V_{DS}}{2} \right) (1 + \lambda V_{DS}) \quad \text{per } V_{DS} \leq V_{GS} - V_T$$

dove $K_N = \mu_N C_{OX}$ e $C_{OX} = \frac{\epsilon_S \epsilon_0}{t_{OX}}$ e $L_1 = L - 2L_D$ e $W_1 = W - 2W_D$.

Le correzioni a L e W , pari rispettivamente a $2L_D$ e a $2W_D$, sono dovute ad errori sistematici del processo (es. sottodiffusioni). L_D e W_D sono indicati nelle model come **LD** e **WD**. La tensione di soglia è espressa come:

$$V_T = V_{T0} + \gamma (\sqrt{\phi - V_{BS}} - \sqrt{\phi}) \quad (\text{A2.2})$$

Questa relazione rappresenta la dipendenza della tensione di soglia dalla tensione di substrato (rispetto al source) dovuta all'effetto body. I vari parametri delle equazioni (A2.1) e (A2.2) sono determinati dai parametri della MODEL:

Parametro delle equazioni	Parametro MODEL	Espressione alternativa valida in assenza di definizione diretta del parametro
C_{OX}	Non Esiste	$\frac{\epsilon_{rSi} \epsilon_0}{t_{OX}}$
K_N	KP	$\mu_N C_{OX}$
t_{OX}	TOX	
μ_N	UO	
λ	LAMBDA (solo LEVEL=1)	
V_{T0}	VTO	espressione in base al drogaggio e altri parametri
γ	GAMMA	espressione in base a drogaggio e altri parametri
ϕ	PHI	espressione in funzione del drogaggio

Nei LEVEL superiori al primo l'espressione della I_{DS} rimane una funzione delle tre tensioni V_{GS} , V_{DS} e V_{BS} ma risulta più complessa e, in particolare, risulta meno discontinuo il passaggio dalla saturazione alla zona triodo e, per modelli più recenti, dalla zona di forte inversione alla zona di sottosoglia. I parametri introdotti nella tabella precedente risultano ancora tutti validi tranne il parametro LAMBDA che non ha più significato nel LEVEL=3.

Nel LEVEL=3 l'espressione della corrente in saturazione è data da:

$$I_{DS} = I_{DSAT} \frac{1}{1 - \frac{\Delta L}{L_1}} \quad (A2.3)$$

dove I_{DSAT} è la corrente al limite della saturazione, L_1 è la lunghezza efficace del canale prima dello strozzamento mentre ΔL è l'accorciamento del canale (valore positivo) dovuto allo strozzamento. ΔL è quindi una funzione crescente di V_{DS} .

Dall'espressione (3) può essere calcolato un valore di λ considerando che, dalla (A2.1) in saturazione:

$$\lambda = \frac{1}{I_{DSAT}} \frac{\partial I_{DS}}{\partial V_{DS}}$$

e pertanto, derivando la (A2.3), se si può approssimare $1 - \Delta L/L_1$ a circa 1, si ottiene:

$$\lambda = \frac{1}{L_1} \frac{\partial \Delta L}{\partial V_{DS}} \quad (A2.4)$$

In genere $\frac{\partial \Delta L}{\partial V_{DS}}$ si può considerare che dipenda solo da $V_{DS} - V_{DSAT}$ e pertanto il valore di

λ diminuisce al crescere della lunghezza di canale. Ciò corrisponde alla realtà sperimentale ed è ben rappresentato da simulazioni effettuate sia con LEVEL=2 sia con LEVEL=3.

I diodi source-substrato (D_{BS}) e drain-substrato (D_{BD}) sono caratterizzati da una corrente di saturazione $I_s = JS \times AS$ per D_{BS} e $I_s = JS \times AD$ per il diodo D_{BD} . Il parametro JS è fornito nelle model, mentre AS e AD sono specificati nella dichiarazione del dispositivo nella netlist.

CAPACITÀ NEL MODELLO DEL MOS

Con riferimento al modello per grandi segnali abbiamo sei capacità, tutte variabili con la tensione a cui sono sottoposte. Pertanto si parlerà di capacità differenziali.

Capacità di gate: C_{GS} C_{GD} C_{GB} . Queste capacità possono essere scritte come:


$$C_{GS} = C_{GS}^{(c)} + C_{GS}^{(s)}; \quad C_{GD} = C_{GD}^{(c)} + C_{GD}^{(s)}; \quad C_{GB} = C_{GB}^{(c)} + C_{GB}^{(s)}$$

dove l'apice (c) indica "capacità di canale" e l'apice (s) indica la capacità dovuta alla sovrapposizione del gate con il rispettivo elettrodo.

Le capacità di sovrapposizione si ottengono dai parametri delle model CGSO, CGDO e CGBO. In particolare:

$$C_{GS}^{(s)} = CGSO \times W; \quad C_{GD}^{(s)} = CGDO \times W; \quad C_{GB}^{(s)} = CGBO \times L$$

La figura seguente mostra il layout di un MOSFET evidenziando l'origine delle capacità di sovrapposizione.



Per quanto riguarda le capacità di canale, esse variano fortemente con la zona di funzionamento del MOS. Nella formulazione più semplice data dal LEVEL 1 si ha:

	$V_{GS} < V_T$	Zona Triodo	Saturazione
$C_{GS}^{(c)}$	0	$1/2 LWC_{OX}$	$2/3 LWC_{OX}$
$C_{GD}^{(c)}$	0	$1/2 LWC_{OX}$	0
$C_{GB}^{(c)}$	LWC_{OX}	0	0

Capacità di giunzione: C_{BS} e C_{BD}

Queste capacità sono date ciascuna da due termini corrispondenti rispettivamente al “fondo” e alle “pareti” delle diffusioni di drain e di source.

$$C_{BS} = \frac{CJ \times AS}{\left(1 - \frac{V_{BS}}{PB}\right)^{MJ}} + \frac{CJSW \times PS}{\left(1 - \frac{V_{BS}}{PB}\right)^{MJSW}}; \quad C_{BD} = \frac{CJ \times AS}{\left(1 - \frac{V_{BD}}{PB}\right)^{MJ}} + \frac{CJSW \times PS}{\left(1 - \frac{V_{BD}}{PB}\right)^{MJSW}}$$


I parametri CJ, CGSW, MJ, MJSW e PB sono indicati nella model, mentre le aree AS e AD e i perimetri PS e PD vengono indicati nella dichiarazione del dispositivo nella netlist.

Resistenze serie: R_D e R_S

Le resistenze serie vengono calcolate attraverso il parametro RSH (resistenza di strato) fornito nella model.

Si ha : $R_D = NRD \times RSH$ e $R_S = NRS \times RSH$. NRD e NRS vengono indicati nella dichiarazione del MOSFET.

Modello a piccolo segnale dei MOSFET




Riassunto di alcuni parametri della MODEL del MOSFET

LEVEL	Tipo di modello selezionato. (modelli universali di Spice: 1,2,3)
VTO	Tensione di soglia per $V_{BS}=0$ e in assenza di effetti di canale corto o stretto.
UO	Mobilità
LD	Riduzione della lunghezza del canale su ciascun lato
WD	Riduzione della larghezza del canale su ciascun lato.
LAMBDA	Pendenza delle caratteristiche in saturazione (non valido per il level 3)
GAMMA	Coefficiente dell'effetto body
PHI	Potenziale di superficie
RSH	Resistenza di strato delle diffusioni di Drain e Source
TOX	Spessore dell'ossido
JS	Densità di corrente di saturazione delle giunzioni tra diffusioni e substrato
CJ	Capacità di giunzione a potenziale nullo (da moltiplicare per l'area della giunz.)
CJSW	Capacità di giunzione a potenziale nullo (da moltiplicare per il perimetro)
MJ	Coefficiente di grading per le giunzioni (relativo al termine di area)
MJSW	Coefficiente di grading per le giunzioni (relativo al termine di perimetro)
PB	Potenziale di contatto delle giunzioni tra diffusioni e substrato
CGSO	Capacità di sovrapposizione gate-source (da moltiplicare per W)
CGDO	Capacità di sovrapposizione gate-drain (da moltiplicare per W)
CGBO	Capacità di sovrapposizione gate-bulk (da moltiplicare per L)

Transistori bipolari (BJT)


Nella figura seguente vengono indicate schematicamente le tipiche sezioni dei dispositivi bipolari per circuiti integrati. Lo scopo della figura è far comprendere l'origine di alcuni elementi, quali, per esempio, le capacità verso il substrato che non sono presenti in un componente discreto (provvisto dei soli tre terminali base emettitore e collettore). Gli schemi fanno riferimento a un processo bipolare planare epitassiale o in generale a processi in cui è presente uno strato epitassiale di tipo n , quali alcuni tipi di BiCMOS e di BCD. I componenti sono sempre racchiusi in sezioni di strato epitassiale, denominate "sacche" (*epi pocket*) isolate in basso dalla giunzione in inversa con il substrato (di tipo p) e lateralmente da diffusioni di di tipo p o da scavi (trench) di isolamento. Le sezioni sono comunque egualmente rappresentative di quei processi BiCMOS nei quali la funzione delle "sacche" di epi è svolta da n -well.



La dichiarazione del transistor bipolare (vedi appendice 1) prevede nella sintassi un nodo di substrato opzionale. Con riferimento alla figura precedente, i transistori NPN (verticale) e PNP laterale necessitano di un terminale di substrato poiché non si può prescindere dalle capacità che il collettore (nel NPN) e la base (nel PNP) hanno verso il substrato. Il transistor PNP di substrato non necessita di un quarto terminale in quanto il substrato coincide fisicamente con il collettore. Come accennato precedentemente i transistori discreti non necessitano di terminale di substrato.

Modello statico del transistor bipolare.

Per il modello statico SPICE utilizza il modello di Ebers-Moll nella versione a trasporto. Questo è matematicamente equivalente al modello a iniezione della figura seguente.



$$IS = \alpha_F I_{ES} = \alpha_R I_{CS}$$

$$BF = \frac{\alpha_F}{1 - \alpha_F} \quad BR = \frac{\alpha_R}{1 - \alpha_R}$$

Sono riportate anche le equazioni che legano i parametri BF, BR e IS di Spice (tipici del modello a trasporto) ai parametri del modello di Ebers-Moll a iniezione. Ricordiamo che le correnti I_F e I_R sono date da:

$$I_F = I_{ES} \left(e^{\frac{V_{BE}}{\eta_F V_T}} - 1 \right); \quad I_R = I_{CS} \left(e^{\frac{V_{BC}}{\eta_R V_T}} - 1 \right);$$

Inoltre sono specificati anche due coefficienti di idealità per i due diodi, ovvero η_F (NF nella MODEL) per il diodo B-E e η_R (NR nella MODEL) per il diodo B-C.

Il modello di Ebers-Moll così riportato non tiene conto dei seguenti fenomeni:

1. Resistenze in serie al collettore (R_{cc}), all'emettitore (R_{ee}) e alla base (R_{bb}).
2. Dipendenza di I_C in zona attiva da V_{BC} (ovvero in pratica da V_{CE}) dovuta all'effetto Early.
3. Variazione del guadagno di corrente β in funzione della corrente di collettore.
4. Effetti di alta iniezione.

Le modifiche dell'Ebers-Moll introdotte per tenere conto dei problemi indicati consistono nel modello di Gummel-Poon statico.

Per quanto riguarda il correttivo per l'effetto dovuto alle resistenze serie abbiamo tre parametri, indicati con RC, RE, RB nella model, che appunto costituiscono tre resistenze inserite in serie rispettivamente a collettore, emettitore e base.

Per quanto riguarda l'effetto Early, vengono introdotti due parametri, VAF e VAR, corrispondenti alle tensioni di Early rispettivamente in zona attiva diretta e zona attiva inversa. A titolo di esempio riportiamo l'equazione della corrente di collettore utilizzata da SPICE per la zona attiva diretta (trascurando le correnti di saturazione della giunzione B-C):

$$I_C = IS \cdot e^{\frac{V_{BE}}{NF \cdot V_T}} \left(1 + \frac{V_{CB}}{VAF} \right)$$

Per quanto riguarda la variazione del β (e quindi di β_F e β_R) richiamiamo il fatto che il guadagno di corrente assume il valore massimo (corrispondente in SPICE a BF e BR) solo per un intervallo di correnti di collettore detta regione delle medie correnti. Per correnti inferiori (regione delle piccole correnti) il guadagno decresce significativamente (scendendo anche sotto l'unità) a causa delle correnti di ricombinazione nella zona di carica spaziale della giunzione polarizzata in diretta. Queste correnti, infatti, contribuiscono alla corrente di base ma non sono utili per la corrente di collettore. In SPICE questo fenomeno è modellato aggiungendo due diodi "di perdita" in parallelo ai diodi della I_F e I_R dell'Ebers Moll. Questi diodi hanno una corrente di saturazione inversa e un coefficiente di idealità (di solito intorno a 2) che costituiscono quattro parametri aggiuntivi di SPICE detti parametri di "emissione".

Per correnti superiori alla zona delle medie correnti si ha la regione delle alte correnti dove il guadagno di corrente scende ancora, questa volta per effetti di alta iniezione. Esula da questa trattazione una descrizione dettagliata dei parametri SPICE corrispondenti a questo fenomeno.

Modello dinamico del transistor bipolare.

Il modello dinamico per il transistor bipolare utilizzato da SPICE è il modello a controllo di carica. La carica risulta immagazzinata nelle due giunzioni dei diodi del modello dell'Ebers Moll e risulta data, per ciascuna delle due giunzioni dalla somma:

$$Q = Q_D + Q_J$$

dove Q_D è la capacità di diffusione dovuta all'accumulo di minoritari nella base e nel corrispondente elettrodo (collettore o emettitore, a seconda di quale dei due diodi si considera), mentre la Q_J è la capacità di giunzione associata alle cariche fisse nella zona di svuotamento.

La carica Q_J viene espressa mediante le capacità non lineari di giunzione. Come per il MOSFET la capacità di giunzione viene modellata con l'espressione:

$$C_J = \frac{C_{j0}}{\left(1 - \frac{V_D}{V_0}\right)^m}$$

dove C_{j0} è la capacità per tensione nulla (sul diodo), V_0 è il potenziale di contatto, m è il coefficiente di grading e V_D la tensione sul diodo. Nella model si trovano i seguenti parametri:

CJE, VJE, MJE, corrispondenti ai parametri C_{j0} , V_0 , e m per il diodo BE;

CJC, VJC, MJC, corrispondenti ai parametri C_{j0} , V_0 , e m per il diodo BC;

Vi sono inoltre i parametri CJS, VJS, MJS, corrispondenti ai parametri C_{j0} , V_0 , e m per la giunzione tra il collettore e il substrato (transistore verticale NPN) o tra la base e il substrato (transistore PNP laterale). T-Spice ha il parametro SUBS che assume valore -1 per struttura laterale e 1 per struttura verticale.

La carica di diffusione Q_D viene invece associata alla corrente mediante due parametri, detti tempi di transito: TF per la giunzione BE, TR per la giunzione BC.

Le cariche Q_{DE} (giunzione BE) e Q_{DC} (giunzione BC) risultano:

$$Q_{DE} = TF \cdot I_S \left(e^{\frac{V_{BE}}{V_T}} - 1 \right)$$
$$Q_{DC} = TR \cdot I_S \left(e^{\frac{V_{BC}}{V_T}} - 1 \right)$$

In zona attiva diretta, dove l'unità può essere trascurata rispetto all'esponenziale si ha:

$$C_{DE} = \frac{\partial Q_{DE}}{\partial V_{BE}} = TF \cdot \frac{I_S}{V_T} e^{\frac{V_{BE}}{V_T}} = TF \cdot \frac{I_C}{V_T} = TF \cdot g_m$$

La C_{DE} così calcolata è la capacità di diffusione del modello linearizzato. Un'espressione analoga si può ricavare per la C_{DC} , la capacità di diffusione della giunzione collettore-base

che in zona attiva diretta è trascurabile. Si osservi che in zona di saturazione entrambe le capacità di diffusione sono significative e preponderanti rispetto alle capacità di giunzione. Si può derivare una semplice approssimazione della frequenza di transizione f_T :

A rigore, in zona attiva diretta:

$$f_T = \frac{g_m}{2\pi(C_{jE} + C_{jC} + C_{DE})}$$

Poichè quando la giunzione BE è in diretta si può ritenere valida l'approssimazione $C_{DE} \gg C_{jE} + C_{jC}$, pertanto:

$$f_T \cong \frac{g_m}{2\pi C_{DE}} \cong \frac{1}{2\pi TF}$$

Questa espressione è utile per ottenere una stima approssimata della frequenza di transizione a partire dal parametro TF della MODEL.